
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

PROJECT: CIRCULAR FASHION ECONOMY 

 

TECHNICAL NOTE No 2 

 

 

THE CGE FASHION MODEL 

 

 

 

 

 

prepared by Arman Mazhikeyev  

and Marian Rizov  

 

 

 

Date: 21 November 2020  

 

 

 

 

 



1. Introduction 
 

This report describes the CGE model explicitly incorporating the fashion sector; the model is 

developed as a part of the Circular Fashion project (BSR). This is a multi-region, multi-sector 

and multi-factor model developed for the purposes of simulation analyses. The model 

includes 5 regions (China, India, USA, EU28 and Rest of the World), 27 sectors and 8 types 

of production factors. Technically (from mathematical viewpoint), the model is a system of 

non-linear equations that incorporates the core concepts of general equilibrium and links 

actual economic data representing key activities within and across the regions and sectors 

defined. 

Software requirements: The model is written in GAMS/IDE (General Algebraic Modelling 

System with Integrated Development Environment) and solved with CONOPT3 solver. The 

software can be downloaded from the official website: www.gams.com. The model will work 

in version 23.5 or any later versions of GAMS/IDE software. However, due to the high 

dimensionality of the models, any user is required to obtain a program license and purchase 

the CONOPT3 solver: https://www.gams.com/sales/pricing_regular/. 

Model files: The main model file is called “fashion.gms” which contains all the codes. To 

run the model, first, the directory file “dir.gpr” needs to be executed. There are other files 

(fashion.log, fashion.lst, fashion.lxi) which are automatically generated/updated when the 

model (first) runs. The log file records all the actions taken, codes run and results generated 

per run. The lst file opens in separate window once the gms file runs and displays the outputs 

and model statistics. The lxi file contains information to navigate through the lst file (this file 

does not need to be displayed). 

 

2. Model structure 

 

The model has a nested structure as shown in Scheme 1. At the very bottom of the structure, 

total labour is decomposed by gender (F0/M0) and by occupation type. There are 8 categories 

of labour occupation which are grouped into high skilled and low skilled labour groups (H/L) 

combined with a Canonical CES function (F1) to form gross expenditure on labour. The 

labour factor then is aggregated with the non-labour factor of production (K) with a Cobb-

Douglas function to form total production factor (Y). To produce final goods, the (total) 

factors are joined with intermediate factors (X) via a Leontief function to obtain total output 

value (Z). Produced goods are either supplied to domestic (D) market and/or exported (E) and 

distributed via a CES transformation function. At destinations, all traded goods (Q) are either 

from domestic suppliers or imported (M), joined via Armington CES consumption function. 

All the goods, depending on the purpose, are consumed by four agents (firms, households, 

government, and investors). One final branch at the top is the utility received by private 

sector from consumption of goods and services given in a form of a Cobb-Douglas utility 

function (U). 

 

http://www.gams.com/
https://www.gams.com/sales/pricing_regular/


Armington CES: 

𝜎=5 

CET: 𝜀 = 4 

Leontief: 𝜇 = 0 

CD: 𝛽=1 

Canonical CES: 𝜑=1.7  

CD: 𝛼=1 

Scheme 1: Model structure 

3. Model code and procedures 

 

In what follows the model code is introduced, step-by-step and the procedures at key stages 

are described.  

 
$Ontext 

The Model developed as a part of Circular Fashion project. 

Developer A Mazhikeyev from University of Lincoln 

Notes: 

- licensed GAMS/IDE version 2.3 or above required to run the model 

- GAMS' CONOPT3 (or MINOS5) required to solve the model 

- GDX data files are required to run the model 

- data parameters are all in LOWER case letters with standard GTAP abbreviations 

- scale parameters are all in LOWER case LATIN letter form 

- calibrated and elasticity parameters are all in lower case GREEK letter form 

- variables are all in UPPER case letter form 

$Offtext 

 

OPTIONS DECIMALS = 6 ; 

OPTION NLP = CONOPT3 ; 

 

Sets and subsets 

Sectors are denoted with i, regions with r, and factors with f. There are 27 sectors in set i, 

5 regions in set r and 8 factors of production in set f. Aliases present some substitute 

notations in use in some cases as appropriate. Exhibit 1 below represents the breakdown of 

sets into subsets. Factors include groups of labour factors (l) and non-labour factors (nl) 

where the former subset consists of high skilled (hs) and low skilled (ls) labour.  

 

Exhibit 1 
 

*----------------------------------------------------------- 

* defining SETS  -------------------------------------- 

*----------------------------------------------------------- 

 

Sets 

     r        regions    /chn, ind, usa, eur, row/ 

     f        factors    /lnd, clk, shw, tch, mng, agw, capital, nrs/ 

     i        industry   /agr, pws, bca, ext, pfd, mnf, chm, rpp, ely, gdt, 

                         wtr, srv, trd, trp, tcr, ocr, swf, mot, mkc, mcr, 

                         mwa, aao, ddf, tdl, mfw, mlh, olp / 

 

*subsets 

 

     l(f)     labour /clk, shw, tch, mng, agw/ 

     hs(l)    high skill labour /tch, mng/ 

     ls(l)    low skill labour /clk, shw, agw/ ; 

 

Alias (i,j),(i,jj),(r,rr),(f,h),(h,k),(r,rrr); 

 

Appendix 1 provides the full definition of each sector, region, and factors in each set or 

subset. Throughout the rest of the report in describing relationships or dimensions of separate 

variables, parameters, and equations, sets are used interchangeably and presented in brackets.  

 

Data files 

Data files come with “gdx” extension as the data file names are consistent with GTAP data 

account abbreviations (see McDonald and Thierfelder, 2004). However, there are some data 



files for labour breakdown by gender and occupation which are not part of the GTAP 

database (vfm_femp and vfm_memp). Short description of each data used is given below in 

Exhibit 2. Technical Note No1 discusses each data account in more detail. 

 

Exhibit 2 
 

*----------------------------------------------------------- 

* defining DATA   -------------------------------------- 

*----------------------------------------------------------- 

 

Parameter 

viam0(r,i,j)             Imported Intermediate input use in production 

vdam0(r,i,j)             Domestic Intermediate input use in production 

vfm0(r,f,j)              Factor use in production 

viaa_viam0(r,i,j)        Tax on imported Intermediate input use in production 

vdaa_vdam0(r,i,j)        Tax on Domestic Intermediate input use in production 

evfa_vfm0(r,f,j)         Tax on Factor use in production 

prodtax0(r,i)            Production tax 

vom0(r,i,j)              Total Domestic Output Value 

vxwd_vxmd0(r,rr,i)       Tax and subsidy on Export 

vst0(i,r)                Transport services associated with international trade 

vxwd0(r,i,rr)            Export 

vdpm0(i,r)               Consumption of domestic commodities by private sector 

vdgm0(i,r)               Consumption of domestic commodities by government 

vdim0(i,r)               Consumption of domestic commodities by investment sector 

viws_vtwr0(r,rr,i)       Import 

vtwr0(r,rr,i)            Transport Services 

vims_viws0(r,rr,i)       Tariffs on Imported Commodities 

vipm0(i,r)               Consumption of imported commodities by private sector 

vigm0(i,r)               Consumption of imported commodities by government 

viim0(i,r)               Consumption of imported commodities by investment sector 

evoa_vdep0(r,f)          Factor earnings minus Capital Depreciation 

ptaxfact0(r,f)           Income Tax 

vdep0(r,f)               Capital Depreciation 

yh0(r)                   Regional income of households 

yg0(r)                   Regional government's income 

save0(r)                 Regional savings 

tmtax0(r,rr)             Import Tariff Revenue 

tetax0(r,rr)             Export Tax Revenue 

istax0(r,i)              Imported Commodity Sales Tax Revenue 

dstax0(r,i)              Domestic Commodity Sales Tax Revenue 

ftax0(r,f)               Factor Use Tax Revenue 

ptax0(r)                 Production Tax Revenue 

dtax0(r)                 Direct Tax Revenue 

vipa_vipm0(i,r)          Sales Tax on Imported Goods Consumption by Private sector 

vdpa_vdpm0(i,r)          Sales Tax on Domestic Goods Consumption by Private sector 

priv_bal0(r)             Private block balancer 

vigm_viga0(i,r)          Sales Tax on Imported Goods Consumption by Government 

sector 

vdgm_vdga0(i,r)          Sales Tax on Domestic Goods Consumption by Government 

sector 

gov_bal0(r)              Government block balancer 

viia_viim0(i,r)          Sales Tax on Imported Goods Consumption by Investment 

sector 

vdia_vdim0(i,r)          Sales Tax on Domestic Goods Consumption by Investment 

sector 

ptaxinv0(r)              Trade balance in Investment goods market 

vtwr_vst0(r)             Trade balance in transport services market 

viws_vtwr_vxwd0(r,rr)    Trade Balance for Goods Market 

inv_bal0(r)              Investment block balancer 

vtwr20(r,rr)             International Transport Services 

vfm_femp0(r,l,i)         Female labour value added 

vfm_memp0(r,l,i)         Male labour value added 

; 

 

 



Loading data 

gdx files are GAMS compatible. Therefore, they can be loaded, opened, viewed and edited in 

GAMS. The user does not need to open them separately. When the main model file is 

executed, it will automatically import data from the gdx files and display it in the lst window 

as shown in Exhibit 3. 

 

Exhibit 3 
 

*----------------------------------------------------------- 

* loading DATA   -------------------------------------- 

*----------------------------------------------------------- 

 

$GDXIN viam0.gdx 

$LOAD viam0 

$GDXIN 

 

$GDXIN vdam0.gdx 

$LOAD vdam0 

$GDXIN 

 

$GDXIN vfm0.gdx 

$LOAD vfm0 

$GDXIN 

 

$GDXIN viaa_viam0.gdx 

$LOAD viaa_viam0 

$GDXIN 

 

$GDXIN vdaa_vdam0.gdx 

$LOAD vdaa_vdam0 

$GDXIN 

 

$GDXIN evfa_vfm0.gdx 

$LOAD evfa_vfm0 

$GDXIN 

 

$GDXIN prodtax0.gdx 

$LOAD prodtax0 

$GDXIN 

 

$GDXIN vom0.gdx 

$LOAD vom0 

$GDXIN 

 

$GDXIN vxwd_vxmd0.gdx 

$LOAD vxwd_vxmd0 

$GDXIN 

 

$GDXIN vst0.gdx 

$LOAD vst0 

$GDXIN 

 

$GDXIN vxwd0.gdx 

$LOAD vxwd0 

$GDXIN 

 

$GDXIN vdpm0.gdx 

$LOAD vdpm0 

$GDXIN 

 

$GDXIN vdgm0.gdx 

$LOAD vdgm0 

$GDXIN 

 

$GDXIN vdim0.gdx 



$LOAD vdim0 

$GDXIN 

 

$GDXIN viws_vtwr0.gdx 

$LOAD viws_vtwr0 

$GDXIN 

 

$GDXIN vtwr0.gdx 

$LOAD vtwr0 

$GDXIN 

 

$GDXIN vims_viws0.gdx 

$LOAD vims_viws0 

$GDXIN 

 

$GDXIN vipm0.gdx 

$LOAD vipm0 

$GDXIN 

 

$GDXIN vigm0.gdx 

$LOAD vigm0 

$GDXIN 

 

$GDXIN viim0.gdx 

$LOAD viim0 

$GDXIN 

 

$GDXIN evoa_vdep0.gdx 

$LOAD evoa_vdep0 

$GDXIN 

 

$GDXIN ptaxfact0.gdx 

$LOAD ptaxfact0 

$GDXIN 

 

$GDXIN vdep0.gdx 

$LOAD vdep0 

$GDXIN 

 

$GDXIN yh0.gdx 

$LOAD yh0 

$GDXIN 

 

$GDXIN yg0.gdx 

$LOAD yg0 

$GDXIN 

 

$GDXIN save0.gdx 

$LOAD save0 

$GDXIN 

 

$GDXIN tmtax0.gdx 

$LOAD tmtax0 

$GDXIN 

 

$GDXIN tetax0.gdx 

$LOAD tetax0 

$GDXIN 

 

$GDXIN istax0.gdx 

$LOAD istax0 

$GDXIN 

 

$GDXIN dstax0.gdx 

$LOAD dstax0 

$GDXIN 

 

$GDXIN ftax0.gdx 



$LOAD ftax0 

$GDXIN 

 

$GDXIN ptax0.gdx 

$LOAD ptax0 

$GDXIN 

 

$GDXIN dtax0.gdx 

$LOAD dtax0 

$GDXIN 

 

$GDXIN vipa_vipm0.gdx 

$LOAD vipa_vipm0 

$GDXIN 

 

$GDXIN vdpa_vdpm0.gdx 

$LOAD vdpa_vdpm0 

$GDXIN 

 

$GDXIN priv_bal0.gdx 

$LOAD priv_bal0 

$GDXIN 

 

$GDXIN vigm_viga0.gdx 

$LOAD vigm_viga0 

$GDXIN 

 

$GDXIN vdgm_vdga0.gdx 

$LOAD vdgm_vdga0 

$GDXIN 

 

$GDXIN gov_bal0.gdx 

$LOAD gov_bal0 

$GDXIN 

 

$GDXIN viia_viim0.gdx 

$LOAD viia_viim0 

$GDXIN 

 

$GDXIN vdia_vdim0.gdx 

$LOAD vdia_vdim0 

$GDXIN 

 

$GDXIN ptaxinv0.gdx 

$LOAD ptaxinv0 

$GDXIN 

 

$GDXIN vtwr_vst0.gdx 

$LOAD vtwr_vst0 

$GDXIN 

 

$GDXIN viws_vtwr_vxwd0.gdx 

$LOAD viws_vtwr_vxwd0 

$GDXIN 

 

$GDXIN inv_bal0.gdx 

$LOAD inv_bal0 

$GDXIN 

 

$GDXIN vtwr20.gdx 

$LOAD vtwr20 

$GDXIN 

 

$GDXIN vfm_femp0.gdx 

$LOAD vfm_femp0 

$GDXIN 

 

$GDXIN vfm_memp0.gdx 



$LOAD vfm_memp0 

$GDXIN 

 

Display viam0,vdam0,vfm0,viaa_viam0,vdaa_vdam0,evfa_vfm0,prodtax0,vom0 

vxwd_vxmd0,vst0,vxwd0,vdpm0,vdgm0,vdim0,viws_vtwr0,vtwr0,vims_viws0,vipm0,vigm0 

viim0,evoa_vdep0,ptaxfact0,vdep0,yh0,yg0,save0,tmtax0,tetax0,istax0,dstax0 

ftax0,ptax0,dtax0,vipa_vipm0,vdpa_vdpm0,priv_bal0,vigm_viga0,vdgm_vdga0 

gov_bal0,viia_viim0,vdia_vdim0,ptaxinv0,vtwr_vst0,viws_vtwr_vxwd0 

inv_bal0,vtwr20,vfm_femp0,vfm_memp0; 

 

Initial parameters 

Loaded data is used to define parametric variables, taxes, tariffs and subsidies, or share 

parameters. Parametric variables listed in Exhibit 4 are fixed form of (endogenous) variables 

used for calibration. They all start with upper letters (e.g., Tl0). Tax and tariff parameters are 

ratios of values of goods at purchase prices and their values given in market prices. Given 

that frictions (i.e., taxes and tariffs) come in tariff-equivalent terms, price parameters are set 

to 1, so without loss of generality to get commodity values traded as given in the data. Price 

parameters start with lower “p” (e.g., pf0) and tax or tariff parameters start with lower “t” 

(e.g., tauz0). All listed parameters have “0” at the end to distinguish them from variables. 

 

Exhibit 4 
 

*----------------------------------------------------------- 

* defining FIXED PARAMETERS -------------------------- 

*----------------------------------------------------------- 

 

Parameter 

Tl0(r,l,i)      total labour 

Fw0(r,l,i)      female workers 

Mw0(r,l,i)      male workers 

H0(i,r)         high skilled labour 

L0(i,r)         low skilled labour 

Y0(j,r)         composite factor 

F0(h,j,r)       the h-th factor input by the j-th firm 

Xp0(i,r)        household consumption of the i-th good 

X0(i,j,r)       intermediate input 

Z0(j,r)         output of the j-th good 

Tz0(j,r)        production tax 

tauz(i,r)       indirect tax rate 

Td0(r)          direct tax 

FF(h,r)         factor endowment of the h-th factor 

taud(r)         direct tax rate 

Tm0(j,r)        import tariff 

M0(i,r)         imports 

taum(i,r)       import tariff rate 

Tr0(i,rr)       transport services 

taut(j,r)       transport margins 

Tp0(i,rr)       sales tax revenue private consumption 

taup(i,rr)      sales tax rate on private consumption 

Xg0(i,r)        government consumption 

Tg0(i,rr)       sales tax revenue government consumption 

taug(i,rr)      sales tax rate on govnt consumption 

Xv0(i,r)        investment demand 

Tv0(i,rr)       sales tax revenue on investment goods 

tauv(i,rr)      sales tax rate investment goods 

E0(i,r)         exports 

Tx0(i,rr)       export tax minus subsidy revenue function 

taux(i,rr)      export tax rate 

reg0(r)         regional income 

Sg0(r)          government saving 

Sf0(r)          foreign saving in US dollars 

sgov(r)         share of govnt savings 

Sp0(r)          private saving 



sinv(r)         share of investment in reg income 

shoh(r)         share of private savings 

D0(i,r)         domestic good 

Q0(i,r)         Armington's composite good 

pf0(h,r)        factor price 

py0(j,r)        composite factor price 

pz0(i,r)        output price 

pq0(i,r)        Armington composite good price 

pm0(i,r)        imported good price 

pe0(i,r)        exported good price 

epsilon0(r)     exchange rate 

pWe0(i,r)       world exported good price 

pWm0(i,r)       world imported good price 

pd0(i,r)        domestic good price; 

 

fw0(r,l,i)= vfm_femp0(r,l,i); 

mw0(r,l,i)= vfm_memp0(r,l,i); 

tl0(r,l,i)= fw0(r,l,i)+mw0(r,l,i); 

h0(i,r) =   sum(hs, fw0(r,hs,i)+ mw0(r,hs,i) ); 

l0(i,r) =   sum(ls, fw0(r,ls,i)+ mw0(r,ls,i) ); 

F0(h,j,r) =vfm0(r,h,j); 

Y0(j,r)   =sum(h, F0(h,j,r)); 

Xp0(i,r)  =vipm0(i,r) +vdpm0(i,r); 

X0(i,j,r) =viam0(r,i,j)+vdam0(r,i,j); 

Z0(j,r)   =Y0(j,r) +sum(i, X0(i,j,r)); 

Tz0(j,r)  

=prodtax0(r,j)+sum(i,viaa_viam0(r,i,j))+sum(i,vdaa_vdam0(r,i,j))+sum(h,evfa_vfm0(r,

h,j)); 

tauz(j,r) =Tz0(j,r)/Z0(j,r); 

FF(h,r)   =evoa_vdep0(r,h)+ptaxfact0(r,h)+vdep0(r,h); 

Td0(r)    =sum(h,ptaxfact0(r,h)); 

taud(r)   =Td0(r)/sum(h, FF(h,r)); 

M0(i,rr)   =sum(r,viws_vtwr0(rr,r,i)); 

Tr0(i,rr)  =sum(r,vtwr0(rr,r,i)); 

Tm0(i,rr)  =sum(r,vims_viws0(rr,r,i)); 

taut(j,r) =Tr0(j,r)/M0(j,r); 

taum(j,r) =Tm0(j,r)/((1+taut(j,r))*M0(j,r)); 

Tp0(i,r)  =vipa_vipm0(i,r) +vdpa_vdpm0(i,r); 

taup(i,r) =Tp0(i,r)/Xp0(i,r); 

Xg0(i,r)  =vigm0(i,r) +vdgm0(i,r); 

Tg0(i,r)  =vigm_viga0(i,r) +vdgm_vdga0(i,r); 

taug(i,r)$(Xg0(i,r)>0) =Tg0(i,r)/Xg0(i,r); 

Xv0(i,r)  =viim0(i,r) +vdim0(i,r); 

Tv0(i,r)  =viia_viim0(i,r) +vdia_vdim0(i,r); 

tauv(i,r)$(Xv0(i,r)>0) =Tv0(i,r)/Xv0(i,r); 

E0(i,r)   =sum(rr,vxwd0(r,i,rr)); 

Tx0(i,r)  = sum(rr, vxwd_vxmd0(r,rr,i)); 

taux(i,r) =Tx0(i,r)/E0(i,r); 

reg0(r)=  (sum(h,FF(h,r)-vdep0(r,h))-Td0(r)) +sum(i,Tm0(i,r))+sum(rr,tetax0(r,rr)) 

           +sum(i, Tp0(i,r)+Tg0(i,r)+Tv0(i,r)+Tz0(i,r)) +Td0(r) ; 

Sf0(r)     = sum(i,M0(i,r)) -sum(i, E0(i,r)) ; 

Sg0(r)    =(save0(r) - Sf0(r))*(yg0(r)/(yh0(r)+yg0(r))); 

Sp0(r)    =(save0(r) - Sf0(r))*(yh0(r)/(yh0(r)+yg0(r))); 

sgov(r)    =(yg0(r))/reg0(r); 

sinv(r) =(save0(r))/reg0(r); 

shoh(r) =(yh0(r))/reg0(r); 

D0(i,r)   =(1+tauz(i,r))*Z0(i,r)+sum(rr, vxwd_vxmd0(r,rr,i))- vst0(i,r) -

sum(rr,vxwd0(r,i,rr)); 

Q0(i,r)   =(Xp0(i,r)+Xg0(i,r)+Xv0(i,r)+sum(j, X0(i,j,r))); 

pf0(h,r) =1; 

py0(j,r) =1; 

pz0(i,r) =1; 

pq0(i,r) =1; 

pm0(i,r) =1; 

pe0(i,r) =1; 

epsilon0(r)=1; 

pWe0(i,r)=1; 

pWm0(i,r)=1; 



pd0(i,r) =1; 

 

Display 

h0,l0,F0,Y0,Xp0,X0,Z0,Tz0,tauz,FF,Td0,taud,M0,Tr0,Tm0,taut,taum,Tp0,taup,Xg0,Tg0,ta

ug 

Xv0,Tv0,tauv,E0,Tx0,taux,reg0,Sf0,Sg0,Sp0,sgov,sinv,shoh,D0,Q0; 

 

Elasticity parameters 

There are three key elasticity parameters exogenously introduced (see Exhibit 5): 

- Sigma: The elasticity of substitution of domestic to foreign goods is denoted with 

sigma (>1). The common rate of the sigma used in the relevant literature (Anderson & 

van Wincoop, 2003; Novy & Chen, 2012) is between 2 and 10. The sigma is fixed to 

5 in this model which is somewhat average level. If sigma increases, consumer goods 

of different origins become more substitutable (i.e., differences become smaller).  

- Psi: It represents the elasticity of transformation of goods from domestic to foreign 

market. We fix psi to 4 as it is shown in theoretical work (Head and Mayer, 2014) that 

psi should be equal to sigma minus 1. The idea is the higher psi, the easier it is to 

transform a good produced in domestic market to export good.  

- Kappa: The level of substitution of high to low skilled labour is given with kappa 

(>0). The higher the kappa, the easier to substitute high skilled for low skilled labour. 

The common rate of the kappa mentioned in Acemoglu and Autor (2011) is 1.7 and 

this rate is in use in this model. 

- Eta, Phi and Xi are used for simplification of expressions. 

 

Exhibit 5 

 
*----------------------------------------------------------- 

* defining ELASTICITY PARAMETERS ----------------- 

*----------------------------------------------------------- 

 

Parameter 

sigma(i)        elasticity of substitution btw dom and imp cons goods 

psi(i)          elasticity of transformation btw exp and dom prod goods 

eta(i)          substitution elasticity parameter dom to imp goods 

phi(i)          transformation elasticity parameter exp to dom goods 

kappa(i)        elasticity of substitution btw high and low skill labour 

xi(i)           substitution elasticity parameter low to high skill; 

 

sigma(i)=5; 

psi(i)  =4; 

kappa(i)=1.7; 

eta(i)  =(sigma(i)-1)/sigma(i); 

phi(i)  =(psi(i)+1)/psi(i); 

xi(i)  =(kappa(i)-1)/kappa(i); 

 

 

Calibrated parameters 

The share and scale parameters of various functions were calibrated to replicate the initial 

benchmark values given by the data. These are listed in Exhibit 6. 

 

Exhibit 6 
 

*----------------------------------------------------------- 

* calibrated PARAMETERS ------------------- 

*----------------------------------------------------------- 

 

Parameter 

wl0(i,r)        wage of low skilled labour 

wh0(i,r)        wage of high skilled labour 



ah0(i,r)         high skill factor-augmenting technology term 

al0(i,r)         low skill factor-augmenting technology term 

thetac(i,r)     scale parameter 

ah(i,r)        rescaled high skill factor-augmenting technology term 

al(i,r)        rescaled low skill factor-augmenting technology term 

alpha(i,r)      share parameter in utility function 

beta(h,j,r)     share parameter in production function 

b(j,r)          scale parameter in production function 

ax(i,j,r)       intermediate input requirement coeff. 

ay(j,r)         composite fact. input req. coeff. 

mu(i,r)         government consumption share 

lambda(i,r)     investment demand share 

ssp(r)          average propensity for private saving 

ssg(r)          average propensity for gov. saving 

deltam(i,r)     share parameter in Armington function 

deltad(i,r)     share parameter in Armington function 

gamma(i,r)      scale parameter in Armington function 

xid(i,r)        share parameter in transformation function 

xie(i,r)        share parameter in transformation function 

theta(i,r)      scale parameter in transformation function; 

 

ah0(i,r)=h0(i,r)**(1-xi(i))/(h0(i,r)**(1-xi(i))+l0(i,r)**(1-xi(i))); 

al0(i,r)=l0(i,r)**(1-xi(i))/(l0(i,r)**(1-xi(i))+h0(i,r)**(1-xi(i))); 

thetac(i,r)=sum(l, F0(l,i,r))/(ah0(i,r)*h0(i,r)**xi(i)  

+al0(i,r)*l0(i,r)**xi(i))**(1/xi(i)); 

ah(i,r)=ah0(i,r)*(thetac(i,r)**xi(i)); 

al(i,r)=al0(i,r)*(thetac(i,r)**xi(i)); 

alpha(i,r)=Xp0(i,r)/sum(j, Xp0(j,r)); 

beta(h,j,r)=F0(h,j,r)/sum(k, F0(k,j,r)); 

b(j,r)    =Y0(j,r)/prod(h, F0(h,j,r)**beta(h,j,r)); 

ax(i,j,r) =X0(i,j,r)/Z0(j,r); 

ay(j,r)   =Y0(j,r)/Z0(j,r); 

mu(i,r)   =Xg0(i,r)/sum(j, Xg0(j,r)); 

lambda(i,r)=Xv0(i,r)/(sum(j,Xv0(j,r))); 

ssp(r)  =(yh0(r)/(yh0(r)+yg0(r))); 

ssg(r)  =(yg0(r)/(yh0(r)+yg0(r))); 

deltam(i,r)=(1+taum(i,r))*((1+taut(i,r))*M0(i,r))**(1-eta(i)) 

        /((1+taum(i,r))*((1+taut(i,r))*M0(i,r))**(1-eta(i)) +D0(i,r)**(1-eta(i))); 

deltad(i,r)=D0(i,r)**(1-eta(i)) 

        /((1+taum(i,r))*((1+taut(i,r))*M0(i,r))**(1-eta(i)) +D0(i,r)**(1-eta(i))); 

gamma(i,r) =Q0(i,r)/(deltam(i,r)*((1+taut(i,r))*M0(i,r))**eta(i) 

        +deltad(i,r)*D0(i,r)**eta(i))**(1/eta(i)); 

xie(i,r)=(vst0(i,r)+E0(i,r)-taux(i,r)*E0(i,r))**(1-phi(i)) 

       /((vst0(i,r)+E0(i,r)-taux(i,r)*E0(i,r))**(1-phi(i))+D0(i,r)**(1-phi(i))); 

xid(i,r)=D0(i,r)**(1-phi(i))/((vst0(i,r)+E0(i,r)-taux(i,r)*E0(i,r))**(1-

phi(i))+D0(i,r)**(1-phi(i))); 

theta(i,r)=((1+tauz(i,r))*Z0(i,r))/(xie(i,r)*(vst0(i,r)+E0(i,r)-taux(i,r)*E0(i,r) 

)**phi(i) 

        +xid(i,r)*D0(i,r)**phi(i))**(1/phi(i)); 

wl0(i,r) = al(i,r)**xi(i)*(  al(i,r)**xi(i) 

       + ah(i,r)**xi(i)*(H0(i,r)/L0(i,r))**xi(i))**(1/(kappa(i)-1)) ; 

wh0(i,r) = ah(i,r)**xi(i)*( ah(i,r)**xi(i) 

       + al(i,r)**xi(i)*(H0(i,r)/L0(i,r))**(-(kappa(i)-1)/kappa(i)))**(1/(kappa(i)-

1)) ; 

 

Display alpha,beta,b,ax,ay,mu,lambda,deltam,deltad,gamma,xie,xid,theta; 

 

*----------------------------------------------------------- 

* list of VARIABLES ------------------- 

*----------------------------------------------------------- 

 

Variable 

Fw(r,l,i)       female workers 

Mw(r,l,i)       male workers 

Totw(l,r)       total workers 

FF(h,r)         factor income 

Hv(i,r)         high skilled labour 

Lv(i,r)         low skilled labour 



wh(i,r)         high skilled labour wage 

wl(i,r)         low skilled labour wage 

Y(j,r)          composite factor of production 

F1(h,j,r)       the h-th factor input by the j-th firm 

Xp(i,r)         household cons. of the i-th good 

X(i,j,r)        intermediate input 

Z(j,r)          output of the j-th good 

pf(h,r)         the h-th factor price 

py(j,r)         composite factor price 

pz(i,r)         supply price of the i-th good 

pq(i,r)         Armington's composite good price 

Tz(j,r)         production tax 

Td(r)           direct tax 

M(i,r)          imports at market prices 

pm(i,r)         import price in local currency 

Tm(i,r)         import tariff revenue 

Tr(i,r)         international transport services 

Tp(i,r)         sales tax on consumption by private sector 

Xg(i,r)         government consumption 

Tg(i,r)         sales tax on consumption by government 

Xv(i,r)         investment demand 

Tv(i,r)         sales tax on consumption by investment 

E(i,r)          exports at world prices 

Tx(i,r)         export tax minus subsidy 

pe(i,r)         export price in local currency 

reg(r)          regional income level 

vdep(r,h)       capital depreciation 

Sg(r)           government saving 

gov_bal(r)      govnt block balancer 

vst(i,r)        export of transport services 

priv_bal(r)     private block balancer 

inv_bal(r)      investment block balancer 

Sp(r)           private saving 

save(r)         overall regional saving 

epsilon(r)      exchange rate 

pWe(i,r)        export price in US dollars 

pWm(i,r)        import price in US dollars 

Q(i,r)          Armington's composite good 

D(i,r)          domestic good 

pd(i,r)         the i-th domestic good price 

bE(r,i,rr)      bilateral export at world prices 

bM(rr,r,i)      bilateral import excluding transport expenses 

UU(r)           utility 

SW              social welfare; 

 

*----------------------------------------------------------- 

* list of EQUATIONS ------------------- 

*----------------------------------------------------------- 

 

Equation 

eqh(i,r)        supply of high skilled workers of both gender 

eql(i,r)        supply of high skilled workers of both gender 

eq_can(i,r)     labour production function (canonical model) 

wleq(i,r)       wage function for low skilled labour 

wheq(i,r)       wage function for high skilled labour 

eqpy(j,r)       composite factor aggregation func. 

eqX(i,j,r)      intermediate demand function 

eqY(j,r)        composite factor demand func. 

eqF(h,j,r)      factor demand function 

eqpzs(j,r)      unit cost function 

eqTz(j,r)       production tax revenue function 

eqTd(r)         direct tax revenue function 

eqTm(i,r)       import tariff revenue function 

eqTr(i,r)       transport margin function 

eqTp(i,r)       sales tax function on private consumption 

eqTg(i,r)       sales tax function on government consumption 

eqTv(i,r)       sales tax function on investment consumption 

eqTx(i,r)       export tax minus subsidy function 



eqreg(r)       regional income function 

eqXg(i,r)       government demand function 

eqXv(i,r)       investment demand function 

eqSp(r)         private saving function 

eqSg(r)         government saving function 

eqsave(r)       regional savings function 

eqXp(i,r)       household demand function 

eqpe(i,r)       world export price equation 

eqpm(i,r)       world import price equation 

eqepsilon(r)    balance of payments 

eqpqs(i,r)      Armington function 

eqM(i,r)        import demand function 

eqD(i,r)        domestic good demand function 

eqpzd(i,r)      transformation function 

eqE(i,r)        export supply function 

eqDs(i,r)       domestic good supply function 

eqpqd(i,r)      market clearing cond. for comp. good 

eqpf(h,r)       factor market clearing condition 

eqpw(i,r,rr)    international price equilibrium 

eqw(i,r,rr)     international quantity equilibrium 

eqw2(i,r,rr)    international quantity equilibrium 

eqE2(i,r)       equity of total and bilateral export 

eqM2(i,rr)      equity of total and bilateral import 

eqUU(r)         utility function 

obj             social welfare function 

eqtotw(l,r) 

eqtothl(r) 

; 

 

4. Defining model equations 
 

Factor-augmenting production 

Following notation in Scheme 1, gross value added (F1) takes the constant elasticity of 

substitution function of high skilled labour (Hv) and low skilled labour (Lv). The substitution 

parameter (xi) is a function of substitution elasticity (kappa>0) between high and low skilled 

labour in production in sector i of r. The kappa is fixed to 1.7 (as reported in Acemoglu and 

Autor, 2011).    
 

eq_can(i,r)..  sum(l, F1(l,i,r)) =e= (ah(i,r)*hv(i,r)**xi(i) 

                       +al(i,r)*lv(i,r)**xi(i))**(1/xi(i)); 

 

If the elasticity level is larger than 1 then high and low skilled labour are gross substitutes. As 

kappa increases Hv and Lv are more substitutable and vice-versa.   

 

Unit wage function for low skill labour is given as a function of marginal value added by the 

low skilled labour:   
 

wleq(i,r)..  wl(i,r) =e= al(i,r)**xi(i)*(  al(i,r)**xi(i) 

             + ah(i,r)**xi(i)*(hv(i,r)/lv(i,r))**xi(i))**(1/(kappa(i)-1)) ; 

 

Unit wage function for high skill labour is given as a function of marginal value added by the 

high skilled labour: 
 

wheq(i,r)..  wh(i,r)=e= ah(i,r)**xi(i)*( ah(i,r)**xi(i) + al(i,r)**xi(i) 

               *(hv(i,r)/lv(i,r))**(-(kappa(i)-1)/kappa(i)))**(1/(kappa(i)-1)) ; 

 

Total number of high and low skilled labour in each r is equal to the sum of total female and 

male high and low skill labour: 
 

eqh(r)..       sum(i,hv(i,r)) =e= sum(i, sum(hs, fw(r,hs,i)+ mw(r,hs,i) )); 



eql(r)..       sum(i,lv(i,r)) =e= sum(i,  sum(ls, fw(r,ls,i)+ mw(r,ls,i) 

)); 

 

Domestic production 

Value added is produced from factor h used in j of r with the Cobb-Douglas production 

technology: 
 

eqpy(j,r)..     Y(j,r)  =e= b(j,r)*prod(h, F1(h,j,r)**beta(h,j,r)); 

 

In gross output production, factor inputs and intermediates are used as complements with the 

Leontief technology. Intermediate input i used by j in r is used in a proportion (ax) of gross 

output volume (Z): 
 

eqX(i,j,r)..    X(i,j,r)=e= ax(i,j,r)*Z(j,r); 

 

Factor input i is a proportion (ay) of gross output volume (Z): 
 

eqY(j,r)..      Y(j,r)  =e= ay(j,r)*Z(j,r); 

eqF(h,j,r)..    F1(h,j,r)=e= beta(h,j,r)*py(j,r)*Y(j,r)/pf(h,r); 

eqpzs(j,r)..    pz(j,r) =e= ay(j,r)*py(j,r) +sum(i, ax(i,j,r)*pq(i,r)); 

 

 

Government behaviour – taxes and tariffs  

Production tax revenue (Tz) depends on fixed production tax rate (tauz), variable output price 

(pz) and output volume (Z):  
 

eqTz(i,r)..     Tz(i,r) =e= tauz(i,r)*pz(i,r)*Z(i,r); 

 

Direct income tax revenue (Tz) is a function of fixed income tax rate (taud), variable factor 

price level (pf) and total factor earnings (FF):  
 

eqTd(r)..       Td(r)   =e= taud(r)*sum(h, pf(h,r)*FF(h,r)); 

 

Import tariff revenue (Tm) is a product of fixed import tariff rate (taum) and transport margin 

(taut), variable import price (pm) and import volume (M):  
 

eqTm(i,r)..     Tm(i,r) =e= taum(i,r)*pm(i,r)*((1+taut(i,r))*M(i,r)); 

 

International transport earnings (Tr) is equal to transport margin rate (taut), variable import 

price (pm) and import volume (M):  
 

eqTr(i,r)..     Tr(i,r) =e= taut(i,r)*pm(i,r)*M(i,r); 

 

Private sales tax revenue (Tp) is a function of fixed sales tax rate applied on purchases by 

households (taup), variable final good price (pq) and final good consumption volume by 

private sector (Xp):  
 

eqTp(i,r)..     Tp(i,r) =e= taup(i,r)*pq(i,r)*Xp(i,r); 

 

Government sales tax revenue (Tg) is a function of fixed sales tax rate applied on purchases 

by government (taug), variable final good price (pq) and final good consumption volume by 

government (Xg):  
 

eqTg(i,r)..     Tg(i,r) =e= taug(i,r)*pq(i,r)*Xg(i,r); 

 

Investment sales tax revenue (Tv) is a function of fixed sales tax rate applied on investment 

goods (tauv), variable final good price (pq) and final good volume used by investment sector 

(Xv):  



 

eqTv(i,r)..     Tv(i,r) =e= tauv(i,r)*pq(i,r)*Xv(i,r); 

 

Export tax revenue (Tx) is formed by fixed export tax minus subsidy rate (taux), variable 

export price (pe) and export volume (E):  
 

eqTx(i,r)..     Tx(i,r) =e= taux(i,r)*pe(i,r)*E(i,r); 

 

Regional income (reg) is equal total tax revenues collected plus total factor earnings minus 

capital depreciation:  
 

eqreg(r)..    reg(r) =e= ( (sum(h,FF(h,r)-vdep(r,h))-Td(r)) 

                    +sum(i,Tm(i,r))+sum(i,Tx(i,r)) +sum(i,Tp(i,r)+Tg(i,r) 

                       +Tv(i,r)+Tz(i,r)) +Td(r) ); 

 

Government consumption (Xg) depends positively on variable share of regional income 

(reg*sgov) minus government savings (Sg) and government sales taxes paid and adversely on 

consumer price level (pq):  
 

 

eqXg(i,r)..     Xg(i,r) =e= mu(i,r)*(sgov(r)*reg(r)+Sg(r)+gov_bal(r) 

                            -sum(j,Tg(j,r))-Sg(r))/pq(i,r); 

 

 

Investment and saving behaviour 

Investment purchases (Xv) depend on regional savings from both private and government 

sector (sinv*reg), foreign savings (Sf), capital depreciation funds (vdep), net transport 

margins minus sales taxes paid on investment goods and adversely on consumer price level 

(pq):  
 

 

eqXv(i,r)..     Xv(i,r) =e= lambda(i,r)*(sum(f,vdep(r,f)) +( sum(j,Tr(j,r))- 

sum(j,vst(j,r)) ) +Sf0(r) 

                 +priv_bal(r) +sinv(r)*reg(r) -sum(j,Tv(j,r))-inv_bal(r))/pq(i,r); 

 

Private savings (Sg) is a share of private savings in regional savings (ssp*save) excluding 

foreign savings (Sf): 
 

eqSp(r)..       Sp(r)   =e= ssp(r)*(save(r) - Sf0(r)); 

 

Government savings (Sg) is a share of government in regional savings (ssg*save) excluding 

foreign savings (Sf): 
 

eqSg(r)..       Sg(r)   =e= ssg(r)*(save(r) - Sf0(r)); 

 

Regional savings (save) is total regional income from taxes minus consumption of goods by 

private (Xp) and government (Xg) sectors including sales taxes paid (Tp+Tg): 
 

 

eqsave(r)..  save(r) =e=   ((sum(h,FF(h,r)-vdep(r,h))-Td(r)) +sum(i,Tm(i,r)) 

             +sum(i,Tx(i,r))+sum(i, Tp(i,r)+Tg(i,r)+Tv(i,r)+Tz(i,r)) +Td(r) 

             -sum(j,Xp(j,r)+Tp(j,r)) -priv_bal(r) -sum(j,Xg(j,r)+Tg(j,r)) 

+gov_bal(r) ); 

 

 

Household consumption 

Private consumption (Xp>0) depends positively on variable share of regional income 

(reg*shoh>0) minus private savings (Sp) and private sales taxes paid and adversely on 

consumer price level (pq):  
 



eqXp(i,r)..     Xp(i,r) =e= alpha(i,r)*(Sp(r)+Td(r)+shoh(r)*reg(r) 

           -sum(j,Tp(j,r))-Sp(r)-Td(r)-priv_bal(r))/pq(i,r); 

 

 

International trade 

Export price (pe) of final goods traded internationally is equal to world export price (pWe) 

after correction for variation in the exchange rate of LCU to USD (level of epsilon is 1 but 

not fixed): 
 

eqpe(i,r)..     pe(i,r) =e= epsilon(r)*pWe(i,r); 

 

Import price (pm) of final goods traded internationally is equal to world import price (pWm) 

after correction for possible variation in the exchange rate of LCU to USD (level of epsilon is 

1 but not fixed): 
 

eqpm(i,r)..     pm(i,r) =e= epsilon(r)*pWm(i,r); 

 

Value of total export (pWe*E) is equal to value of total import (pWm*M) for each region 

plus international transfer of savings (Sf) to the balance of payments: 
 

eqepsilon(r)..  sum(i, pWe(i,r)*E(i,r)) +Sf0(r)    =e= sum(i, pWm(i,r)*M(i,r)); 

 

 

Armington function 

Composite goods (Q) consumed by the agent in r are either of domestic (D) or imported (M) 

origin or both. This relationship is described by the CES (Armington) function where trade 

parameter (eta>0) plays a key role. eta is a function of elasticity of substitution (sigma>1) of 

domestic to imported goods. sigma is fixed to 5 (which is somewhat average; the exact level 

is unknown but it is somewhere between 2 to 10).    
eqpqs(i,r)..    Q(i,r)  =e= gamma(i,r)*(deltam(i,r)*((1+taut(i,r))*M(i,r))**eta(i) 

                            +deltad(i,r)*D(i,r)**eta(i))**(1/eta(i)); 

 

The rationale behind this is the higher the value of sigma the larger the substitution between 

domestic and imported goods; the relative prices of the two matters. This implies if imported 

good price is higher than the domestic good price (possibly because transport costs and 

import tariffs apply on imported good price) then consumer would consume more of the 

domestic good rather than the more expensive imported one. 

 

Demand for imported goods increases with the positive change in either trade, import share, 

scale parameter or domestic good price. An increase in import price (or either increase in 

import tariff level or international transport cost) reduces the demand for imports: 
 

eqM(i,r)..      ((1+taut(i,r))*M(i,r)) =e= (gamma(i,r)**eta(i)*deltam(i,r)*pq(i,r) 

                            /((1+taum(i,r))*pm(i,r)))**(1/(1-eta(i)))  *Q(i,r); 

 

Demand for domestic goods is positively correlated with the positive change in either trade, 

domestic share, scale parameter or composite good price. An increase in domestic good price 

reduces the demand for domestic good: 
 

eqD(i,r)..      D(i,r)  =e= (gamma(i,r)**eta(i)*deltad(i,r)*pq(i,r) 

                            /pd(i,r))**(1/(1-eta(i)))*Q(i,r); 

 

Transformation function 

Domestic output (Z) is either supplied to domestic market (D) or exported to other markets 

(E) or to sold to both markets at the same time. This relationship is modelled with the 

Constant Elasticity of Transformation (CET) function. In this function, transformation of 



goods from domestic to foreign market depends on elasticity of transformation (psi>1). psi is 

fixed to 4 (because in the relevant literature psi is equal to sigma-1).    
 

eqpzd(i,r)..    (1+tauz(i,r))*Z(i,r)  =e= theta(i,r)*(xie(i,r)*(vst(i,r)+E(i,r)-

taux(i,r)*E(i,r) )**phi(i) 

                            +xid(i,r)*D(i,r)**phi(i))**(1/phi(i)); 

 

Since CET is mirror reflection of Armington CES, the interpretation of response to the 

change of transformation elasticity is similar. The larger the psi the greater the transformation 

between domestic and exported good. If the world price for the good becomes higher than the 

domestic good price then more of the good is exported rather than supplied domestically 

(unless some restrictions are imposed by the regulator). 

 

Supply to export markets increases as either transformation elasticity, export share, scale 

parameter or output price grows. However, an increase in export price (or in export tax) 

reduces the demand: 
 

eqE(i,r)..      (vst(i,r)+E(i,r)-taux(i,r)*E(i,r)) =e= (theta(i,r)**phi(i)*xie(i,r) 

                            *pz(i,r)/pe(i,r))**(1/(1-

phi(i)))*((1+tauz(i,r))*Z(i,r)); 

 

Supply to domestic market increases as either transformation elasticity, domestic supply 

share, supply scale parameter or output price increases. However, an increase in domestic 

price reduces the demand: 
 

eqDs(i,r)..     D(i,r)  =e= (theta(i,r)**phi(i)*xid(i,r) 

                            *pz(i,r)/pd(i,r))**(1/(1-

phi(i)))*((1+tauz(i,r))*Z(i,r)); 

 

 

Market clearing condition 

For final goods, total supply of composite good i in r is equal to total consumption by all 

agents in the same r: 

 
eqpqd(i,r)..    Q(i,r)  =e= Xp(i,r)+Xg(i,r)+Xv(i,r)+sum(j, X(i,j,r)); 

 

For factor inputs, total use of a factor in production is equal and fixed to the total available 

factor h in r: 
 

eqpf(h,r)..     FF(h,r) =e= sum(j, F1(h,j,r)); 

 

In labour market, total use of high (or low) skilled labour by occupation type is equal to total 

number of available female and male labour by occupation type in r: 
 

eqtotw(l,r)..   totw(l,r) =e= sum(i,  fw(r,l,i)+ mw(r,l,i) ); 

eqtothl(r)..    sum(l,totw(l,r)) =e= sum(i,hv(i,r)) +sum(i,lv(i,r)); 

 

International market clearing condition 

This imposes the equilibrium of world export price and world import price for each i: 
 
eqpw(i,r,rr)..  (pbE(r,i,rr) -pbM(rr,r,i))$(ord(r) ne ord(rr)) =e= 0; 

 

The equilibrium of total exports plus foreign savings and total imports is: 
 

eqqw(i,r,rr)..  (bE(r,i,rr) -bM(rr,r,i))$(ord(r) ne ord(rr)) =e= 0; 

 

 

Total export of i from r is equal to sum of bilateral export of i to all rr destinations: 



 

eqpEbE(i,r)..    E(i,r)*pe(i,r)   =e= sum(rr,bE(r,i,rr)*pbE(r,i,rr)); 

 

Total import of i from rr is equal to sum of bilateral import of i to all rr origins: 
 

eqpMbM(i,rr)..   M(i,rr)*pm(i,rr)   =e= sum(r,bM(rr,r,i)*pbM(rr,r,i)); 

 

Fictitious objective function 

Utility given in the form of Cobb-Douglas function is the product of private consumption 

powered by the share of composite good in the consumption basket: 
 

parameter  UU0(r); UU0(r)   = prod(i, Xp0(i,r)**alpha(i,r)); 

 

The social welfare (SW) is given as the sum of utility (UU) received by all r:  
 

eqUU(r)..       UU(r)   =e= prod(i, Xp(i,r)**alpha(i,r)); 

obj..           SW      =e= 1; 

 

*----------------------------------------------------------- 

* setting initial level of variables ------------------- 

*----------------------------------------------------------- 

 

FF.fx(h,r) =FF0(h,r) ; 

Totw.fx(l,r) =Totw0(l,r); 

Fw.l(r,l,i) =Fw0(r,l,i); 

Mw.l(r,l,i) =Mw0(r,l,i); 

Hv.l(i,r) =H0(i,r); 

Lv.l(i,r) =L0(i,r); 

wl.l(i,r) = wl0(i,r); 

wh.l(i,r) = wh0(i,r); 

Xp.l(i,r) =Xp0(i,r); 

Y.l(j,r)  =Y0(j,r); 

F1.l(h,j,r)=F0(h,j,r); 

X.l(i,j,r)=X0(i,j,r); 

Z.l(j,r)  =Z0(j,r); 

pf.l(h,r) =1; 

py.l(j,r) =1; 

pz.l(i,r) =1; 

pq.l(i,r) =1; 

Tz.l(i,r) =Tz0(i,r); 

Td.l(r)   =Td0(r); 

M.l(i,r)  =M0(i,r); 

pm.l(i,r) =1; 

Tm.l(i,r) =Tm0(i,r); 

Tr.l(i,r) =Tr0(i,r); 

Tp.l(i,r) =Tp0(i,r); 

Xg.l(i,r) =Xg0(i,r); 

Tg.l(i,r) =Tg0(i,r); 

Xv.l(i,r) =Xv0(i,r); 

Tv.l(i,r) =Tv0(i,r); 

E.l(i,r)  =E0(i,r); 

Tx.l(i,r) =Tx0(i,r); 

reg.l(r)=reg0(r); 

vdep.l(r,h)=vdep0(r,h); 

Sg.l(r)   =Sg0(r); 

gov_bal.fx(r)=gov_bal0(r); 

vst.fx(i,r)=vst0(i,r); 

priv_bal.fx(r)=priv_bal0(r); 

inv_bal.fx(r)=inv_bal0(r); 

Sp.l(r)   =Sp0(r); 

save.l(r)=save0(r); 

epsilon.fx(r)=1; 

pe.l(i,r) =1; 

pWe.fx(i,r)=1; 

pWm.fx(i,r)=1; 

D.l(i,r)  =D0(i,r); 



Q.l(i,r)  =Q0(i,r); 

pd.fx(i,r) =1; 

bE.l(r,i,rr) =vxwd0(r,i,rr); 

bM.l(rr,r,i) =viws_vtwr0(rr,r,i); 

 

*----------------------------------------------------------- 

* setting boundaries of variables ------------------- 

*----------------------------------------------------------- 

 

Fw.lo(r,l,i) =0.000000; 

Mw.lo(r,l,i) =0.000000; 

Hv.lo(i,r) = 0.000001; 

Lv.lo(i,r) = 0.000001; 

Y.lo(j,r) =0.00000; 

F1.lo(h,j,r)=0.0000; 

X.lo(i,j,r)=0.00000; 

Z.lo(j,r) =0.00000; 

pf.lo(h,r)=0.000001; 

py.lo(j,r)=0.000001; 

pz.lo(i,r)=0.000001; 

pq.lo(i,r)=0.000001; 

*Tz.lo(i,r) =0.0000; 

*Td.lo(r)   =0.00001; 

M.lo(i,r) =0.00000; 

pm.lo(i,r)=0.000001; 

*Tm.lo(i,r) =0.0000; 

Tr.lo(i,r) =0.0000; 

*Tp.lo(i,r) =0.0000; 

*Tg.lo(i,r) =0.0000; 

*Tv.lo(i,r) =0.0000; 

Xp.lo(i,r)=0.00000; 

Xg.lo(i,r)=0.00000; 

*Xv.lo(i,r)=0.00000; 

E.lo(i,r) =0.00000; 

*Tx.lo(i,r) =0.0000; 

*reg.lo(r)=0.000001; 

*vdep.lo(r,h)=0.000001; 

*Sg.lo(r)   =0.00001; 

*gov_bal.lo(r)=0.000001; 

Sp.lo(r)   =0.00000; 

*epsilon.lo(r)=0.00001; 

*priv_bal.lo(r)=0.000001; 

*inv_bal.lo(r)=0.000001; 

pe.lo(i,r)=0.000001; 

save.lo(r)=0.00000; 

pWe.lo(i,r)=0.000001; 

pWm.lo(i,r)=0.000001; 

D.lo(i,r) =0.00000; 

pd.lo(i,r)=0.000001; 

Q.lo(i,r) =0.0000; 

*vst.lo(i,r)=0.0000001; 

 

 

5. Assigning numeraire and solving the model 
The model is solved as non-linear system by maximizing social welfare (SW). 

 
pf.fx("nrs",r)=1; 

*epsilon.fx("usa")=1; 

* option iterlim =4; 

Model twocge /all/; 

Solve twocge maximizing SW using nlp; 

 

This is the end of the model code. Further code lines below would allow us to run some 

scenarios (i.e. shocks) and to calculate percentage changes in main variables of interest in the 

model. 
 



6. Shocking the model 
 

This section introduces a list of various shocks used to test the applicability of the model in 

sense if it returns feasible optimal solutions for the given shocks. For example, we assume 

that all import tariffs are removed in international trade of good i in all regions r. In Exhibit 7 

this is the statement taum=0.0 which is followed by solve statement at the end. We can also 

add option bratio=1 to speed up the process and find a satisfying solution; this is an option 

and it can be eliminated. This second solution of the model can be compared with the one 

above which solves the model for baseline data values; second solution solves the model 

(again) considering a change in (removal of) the tariffs. In the second run of the model, all 

the equations pf the system that contain taum will be distorted (i.e. the right-hand side now is 

unequal to the left-hand side) and thus (previous) optimal solution is infeasible. The model 

will go through rounds of small changes where GAMS software attempts to equate the 

equations containing taum by adjusting values of other ‘free’ variables (e.g., import prices 

and values, export prices and values, import revenues, etc.) in those equations. Then, other 

equations that also contain import prices and values, export prices and values, import 

revenues, etc. cause another set of distortions in other parts of the model (e.g., employment, 

production, consumption, etc.) as these cause a wave of disturbed equations. Behavioural 

equations (shown in the Scheme 1) will be triggered and all types of agents from all regions 

start adjusting their assigned functions (e.g., consumers are consuming goods but now 

imported goods are cheaper; firms are exporting and importing goods as import barriers are 

removed; governments are collecting taxes but now no revenue comes from import tariffs, 

etc.). The model will continue iterating and repeating small adjustments in free variables until 

all equations become equal again and all conditions are met, and constraints satisfied. Then a 

report that “feasible and optimal solution” has been reached will be produced but with the 

new solution all variables have slightly changed and differ from their baseline values/levels.  

In a similar manner, we can try using other types of shocks listed below e.g., (tauz is 

production tax, taux is export tax, taut is transport margin, etc.); a full definition of each 

parameter or fixed variable from list below is given above, in Exhibit 4. To implement a 

shock, one need to remove the * sign in front of the relevant statement. It is possible to use 

several shocks at the same time and obtain the mixed effect of the shocks considered.    

 

Exhibit 7 
*----------------------------------------------------------- 

* shocking THE MODEL ------------------- 

*----------------------------------------------------------- 

 

taum(i,r)=0.0; 

*tauz(i,r)=0.0; 

*taux(i,r)=0.0; 

*taut(i,r)=0.0; 

*taup(i,r)=0.0; 

*tauv(i,r)=0.0; 

 

*pf.fx(h,r) =1.1; 

*py.fx(j,r) =1.1; 

*pz.fx(i,r) =1.1; 

*pq.fx(i,r) =1.1; 

*pm.fx(i,r) =1.1; 

*pe.fx(i,r) =1.1; 

*pd.fx(i,r) =1.1; 

 

*sigma(i)=6; 



*eta(i)  =(sigma(i)-1)/sigma(i); 

 

*psi(i)  =5; 

*phi(i)  =(psi(i)+1)/psi(i); 

 

*sigmac(i)=1.7*1.01; 

*phic(i)  =(sigmac(i)-1)/sigmac(i); 

*Totw.l(l,r) =Totw0(l,r); 

*FF.l(h,r) =FF0(h,r) ; 

 

 

*ah(i,r)=1.01*ah0(i,r)*(thetac(i,r)**phic(i)); 

*al(i,r)=1.01*al0(i,r)*(thetac(i,r)**phic(i)); 

 

 

*Fw.fx(r,l,i) =Fw0(r,l,i)*1.05; 

*Mw.l(r,l,i) =Mw0(r,l,i); 

*Totw.l(l,r) =Totw0(l,r); 

*FF.l(h,r) =FF0(h,r) ; 

 

*Hv.fx(i,r) =H0(i,r)*1.01; 

*Lv.l(i,r) =L0(i,r); 

*Totw.l(l,r) =Totw0(l,r); 

*FF.l(h,r) =FF0(h,r) ; 

 

option bratio=1; 

Solve twocge maximizing SW using nlp; 

 

The following parameters calculate the % changes in variables relative to their benchmark 

values.  
 

Parameter 

dY(j,r),dF1(h,j,r),dXp(i,r),dX(i,j,r),dX0(i,j,r),dZ(j,r),dpf(h,r),dpy(j,r),dpz(i,r) 

dpq(i,r),dTz(j,r),dTd(r),dTd0(r),dM(i,r),dpm(i,r),dTm(i,r),dTr(i,r),dTp(i,r),dXg(i,

r) 

dTg(i,r),dXv(i,r),dTv(i,r),dE(i,r),dTx(i,r),dpe(i,r),dreg(r),dSg(r),dgov_bal(r) 

dSp(r),depsilon(r),dpriv_bal(r),dinv_bal(r),dsave(r),dvdep(r,h),dpWe(i,r),dpWm(i,r) 

dD(i,r),dpd(i,r),dQ(i,r),dvst(i,r),dUU(r),dSW,dbE(r,i,rr),dbM(rr,r,i); 

 

dY(j,r) =(Y.l(j,r)  /Y0(j,r)  -1)*100; 

dF1(h,j,r)$(F0(h,j,r)>0)=(F1.l(h,j,r)/F0(h,j,r)-1)*100; 

dXp(i,r)$(Xp0(i,r)>0) =(Xp.l(i,r) /Xp0(i,r) -1)*100; 

dX0(i,j,r)$(X0(i,j,r)>0)=(X.l(i,j,r)/X0(i,j,r)-1)*100; 

dX(i,j,r)$(dX0(i,j,r)>=0)=dX0(i,j,r); 

dZ(j,r)=(Z.l(j,r)  /Z0(j,r)  -1)*100; 

dpf(h,r)=(pf.l(h,r) /1 -1)*100; 

dpy(j,r)=(py.l(j,r) /1 -1)*100; 

dpz(i,r)=(pz.l(i,r) /1 -1)*100; 

dpq(i,r)=(pq.l(i,r) /1 -1)*100; 

dTz(j,r)=(Tz.l(j,r) /Tz0(j,r) -1)*100; 

dTd0(r) =(Td.l(r)   /Td0(r)   -1)*100; 

dTd(r)$(dTd0(r)>0.0000001) =dTd0(r); 

dM(i,r)=(M.l(i,r)  /M0(i,r)  -1)*100; 

dpm(i,r) =(pm.l(i,r) /1 -1)*100; 

dTm(i,r)$(Tm0(i,r)>0)=(Tm.l(i,r) /Tm0(i,r) -1)*100; 

dTr(i,r)$(Tr0(i,r)>0)=(Tr.l(i,r) /Tr0(i,r) -1)*100; 

dTp(i,r)$(Tp0(i,r)>0) =(Tp.l(i,r) /Tp0(i,r) -1)*100; 

dXg(i,r)$(Xg0(i,r)>0) =(Xg.l(i,r) /Xg0(i,r) -1)*100; 

dTg(i,r)$(Tg0(i,r)>0)=(Tg.l(i,r) /Tg0(i,r) -1)*100; 

dXv(i,r)$(Xv0(i,r)>0) =(Xv.l(i,r) /Xv0(i,r) -1)*100; 

dTv(i,r)$(Tv0(i,r)>0)=(Tv.l(i,r) /Tv0(i,r) -1)*100; 

dE(i,r)=(E.l(i,r)  /E0(i,r)  -1)*100; 

dTx(i,r)$(Tx0(i,r)>0)=(Tx.l(i,r) /Tx0(i,r) -1)*100; 

dpe(i,r) =(pe.l(i,r) /1 -1)*100; 



dreg(r) =(reg.l(r) /reg0(r) -1)*100; 

dSg(r) =(Sg.l(r)   /Sg0(r)   -1)*100; 

dgov_bal(r)=(gov_bal.l(r) /gov_bal0(r) -1)*100; 

dSp(r) =(Sp.l(r)   /Sp0(r)   -1)*100; 

depsilon(r)=(epsilon.l(r)/1 -1)*100; 

dpriv_bal(r)=(priv_bal.l(r) /priv_bal0(r) -1)*100; 

dinv_bal(r) =(inv_bal.l(r) /inv_bal0(r) -1)*100; 

dsave(r)  =(save.l(r) /save0(r) -1)*100; 

dvdep(r,h)$(vdep0(r,h)>0) =(vdep.l(r,h) /vdep0(r,h) -1)*100; 

dpWe(i,r) =(pWe.l(i,r)/1 -1)*100; 

dpWm(i,r) =(pWm.l(i,r)/1 -1)*100; 

dD(i,r)  =(D.l(i,r)  /D0(i,r)  -1)*100; 

dpd(i,r)  =(pd.l(i,r) /1 -1)*100; 

dQ(i,r) =(Q.l(i,r)  /Q0(i,r)  -1)*100; 

dvst(i,r)$(vst0(i,r)>0) =(vst.l(i,r)  /vst0(i,r)  -1)*100; 

dbE(r,i,rr)$(vxwd0(r,i,rr)>0)   =   (bE.l(r,i,rr)/vxwd0(r,i,rr) -1)*100; 

dbM(rr,r,i)$(viws_vtwr0(rr,r,i)>0) =(bM.l(rr,r,i) /viws_vtwr0(rr,r,i) -1)*100; 

dUU(r)   =(UU.l(r)  /UU0(r)  -1)*100; 

 

Display dY,dF1,dXp,dX,dZ,dpf,dpy,dpz,dpq,dTz,dTd,dM,dpm,dTm,dTr,dTp,dXg,dTg,dXv 

dTv,dE,dTx,dpe,dreg,dSg,dgov_bal,dSp,depsilon,dpriv_bal,dinv_bal,dsave,dvdep 

dpWe,dpWm,dD,dpd,dQ,dvst,dUU; 

